content: tell application id "com.apple.mail"
  	set theSelectedMessages to selection
  	set the selected_message to item 1 of the theSelectedMessages
  	set message_id to the message id of the selected_message
  	set message_title to the subject of the selected_message
  end tell
  
  -- http://harvey.nu/applescript_url_encode_routine.html
  on urlencode(theText)
  	set theTextEnc to ""
  	repeat with eachChar in characters of theText
  		set useChar to eachChar
  		set eachCharNum to ASCII number of eachChar
  		if eachCharNum = 32 then
  			set useChar to "+"
  		else if (eachCharNum is not equal 42) and (eachCharNum is not equal 95) and (eachCharNum < 45 or eachCharNum > 46) and (eachCharNum < 48 or eachCharNum > 57) and (eachCharNum < 65 or eachCharNum > 90) and (eachCharNum < 97 or eachCharNum > 122) then
  			set firstDig to round (eachCharNum / 16) rounding down
  			set secondDig to eachCharNum mod 16
  			if firstDig > 9 then
  				set aNum to firstDig + 55
  				set firstDig to ASCII character aNum
  			end if
  			if secondDig > 9 then
  				set aNum to secondDig + 55
  				set secondDig to ASCII character aNum
  			end if
  			set numHex to ("%" & (firstDig as string) & (secondDig as string)) as string
  			set useChar to numHex
  		end if
  		set theTextEnc to theTextEnc & useChar as string
  	end repeat
  	return theTextEnc
  end urlencode
  
  return {link: "message://" & urlencode("<" & message_id & ">"), title:message_title}